
1. (12) Determine the value of the variable “count” displayed by each of the three
MsgBox procedures below:

Private Sub Command1_Click()
    Dim n As Integer
    Dim k As Integer
    Dim count As Integer

    For n = 1 To 5
        For k = n To 3
            count = count + 1
        Next k
    Next n
    MsgBox count ‘ What does count equal?

    count = 0
    For n = 1 To 5
        For k = 1 To 5 Step n
            count = count + 1
        Next k
    Next n
    MsgBox count ‘ What does count equal?

    count = 0
    k = 1
    For n = 1 To 5
        Do While k < 5
            If k = 3 Then
                Exit Do
            End If
            k = k + n
            count = count + 1
        Loop
        k = n
    Next n
    MsgBox count ‘ What does count equal?

End Sub

2. (13) You are given the following type definition:

Private Type Stats
    Mean As Double
    StdDev As Double
End Type

You are also given the following click event:

Private Sub cmdCompute_Click()
    Dim a(1 To 50) As Integer
    Dim arrayStats As Stats
    Dim i As Integer

    For i = 1 To UBound(a)
        a(i) = Int(Rnd() * 1000) ‘fill a() with random integers
    Next i

    doStats a(), arrayStats

    MsgBox "Mean = " & arrayStats.Mean
    MsgBox "Standard Deviation = " & arrayStats.StdDev

End Sub

You are required to write the complete procedure “doStats” referenced in the code
above. This procedure is passed an integer array and a variable of the type Stats. The
procedure should compute and store the mean and standard deviation of the numbers
in the array in the “arrayStats” variable. Assume that the array is always full, e.g.,
cells 1 to 50 all store integers.

The formulas for computing the mean and standard deviation are:

mean = (∑ X) / n, and standard deviation = √ ( ∑ X2 ) - ( ∑ X )2 / n

n - 1

where X represents the values in the array.

3. (12) You have the following array named Sales that stores the annual sales in four
regions for the time periods 1995 to 1998.

1 2 3 4

1995

Year 1996

1997

1998

RegionSales array

Assume that sales data has been stored in the array. Write a code segment that
computes the four annual sales figures across the four regions. Each of these annual
totals should be stored in another one-dimensional array named annSales that has
been declared as follows:

Dim annSales(1995 To 1998) As Currency

4. (12) Given the code below, what will be the values stored in the array “a” when the
second nested For … Next loop finishes executing (before the click event
terminates)?

Private Sub Command1_Click()
    Dim a(1 To 4, 1 To 4) As Integer
    Dim i As Integer, j As Integer
    Dim k As Integer

    For i = 1 To 4
        For j = 1 To 4
            k = k + 1
            a(i, j) = k
        Next j
    Next i

    For i = 1 To 4
        For j = 1 To 4
            If i <> j Then
                a(i, j) = a(j, i)
            End If
        Next j
    Next i
    ‘ What is stored in a() here?
End Sub

5. (12) You have a database named “Ages.mdb” that includes a single table named
“ageList”. A portion of this table is shown below:

Write the click event for a command button that computes and displays (via a
Form.Print statement) each age and its frequency. For example, you can see that the
age “23” occurs 4 times in the data above so your code should report that the age 23
has a frequency of 4. Do not print ages that have a frequency of zero. Also assume
that ages fall in the range 1-99 inclusive. Do not make any assumptions about the
number of records in the table.

In writing your solution, assume that a data control named “datAgeList” exists on the
form and has its DatabaseName property set to “Ages.mdb” and its RecordSource
property set to “ageList”.

6. (15) Indicate whether the following statements are True(T) or False(F).

T  or  F By using an SQL Select statement, it is possible to access data in two or
more relational databases using a single data control.

T  or  F A Visual Basic Recordset is always equivalent to a relational database table.

T  or  F You may not always be confident with the accuracy of a Recordset’s
RecordCount property.

T  or  F The ReDim statement is used to redimension an array at run time and it
always retains the values in cells that existed before the dimensions were
changed.

T  or  F The FlexGrid control is strictly an “output only” control. That is, there is no
way for the user to interact with the grid or its cells.

7. (12) If the current record in a Recordset points to the last record and you execute the
Recordset’s MoveNext method, the EOF property is set to True but no error is
generated. However, if you execute the MoveNext method again, an error occurs.
Explain why this is the case.



IS 320 – Winter 1999 Page 5
Exam 2

8. (12) Assume you have a numeric array named “a()” that stores integers and you want
to sort this array into ascending order. Instead of writing a sort procedure, you decide
to take advantage of the list control’s “Sorted” property (set to True). You come up
with the following code segment:

    Dim a(1 To 20) As Integer
    Dim k As Integer

    ‘ the array is populated with unsorted integers – code not shown

    For k = 1 To 20
        List1.AddItem a(k) ‘ put array elements into list (with Sorted set to true)
    Next k

    For k = 1 To 20
        a(k) = List1.List(k - 1) ‘ copy the sorted list back to the array elements
    Next k

    ‘ the array is now sorted

Unfortunately, this strategy does not work.

a.  Explain why it does not work.
b.  Suggest a way to make it work without abandoning the strategy.

1. The three values of the variable “count” are 6,  13,  and  5.

2. An example version of the “doStats” procedure is:

Private Sub doStats(theArray() As Integer, theStats As Stats)
    Dim sumX As Double
    Dim sumX2 As Double
    Dim i As Integer
    Dim n As Integer

    n = UBound(theArray)
    For i = 1 To n
        sumX = sumX + theArray(i)
        sumX2 = sumX2 + theArray(i) ^ 2
    Next i

    theStats.Mean = sumX / n
    theStats.StdDev = Sqr((sumX2 - (sumX ^ 2) / n) / (n - 1))

End Sub

3. The following code segment computes the four annual sales figures:

   For y = 1995 To 1998
        For r = 1 To 4
            annSales(y) = annSales(y) + sales(y, r)
        Next r
    Next y

4. The values stored in the “a()” array are:

a()
1 2 3 4

1 1 5 9 13
2 5 6 10 14
3 9 10 11 15
4 13 14 15 16

5. A sample click event that computes and displays the age frequencies (excluding those
ages with a frequency of zero) is:

Private Sub Command1_Click()
    Dim freq(1 To 100) As Integer
    Dim age As Integer, i As Integer

    Do While Not datAgeList.Recordset.EOF
        age = datAgeList.Recordset("age")
        freq(age) = freq(age) + 1
        datAgeList.Recordset.MoveNext
    Loop

    For i = 1 To 100
        If freq(i) <> 0 Then
            Form1.Print i, freq(i)
        End If
    Next i

End Sub

6. The correct answers to the True/False questions are:

False – You can access two or more tables within the same database, but not two or
more databases.

False – The Recordset can be a single table or the result of an SQL Select statement
where many tables are joined.

True – The RecordCount property may not be accurate until you do a MoveLast in
the RecordSet or MoveNext until you come to the last record.

False – Values are lost using a ReDim unless the Preserve option is used.

False – While it is true that the user cannot enter text into a FlexGrid cell, the user
can click on a cell and make it active. This action can be detected with a
Click event and cause actions to be performed (such as was done with
customer information in Assignment 7).

7. With the first MoveNext, the current record pointer is pointing to a legitimate record
(the last record) before the move occurs. In the second MoveNext, the record pointer
is not pointing to anything before the move occurs and thus generates the error.

8. The code fails because a list box stores data as String type. Thus you get a “String”
sort where, for example,  “11” comes before “2”. To solve the problem, force leading
zeros to be placed in front of the strings written to the list box as follows:

For k = 1 To 20
    List1.AddItem Format$(a(k), "00000")
Next k

This will cause “00002” to come before “00011”.


